Data Report No. 30

HIG-75-19

COMPUTER PROGRAM PROJECTING FICTITIOUS LONGITUDE AND LATITUDE SYSTEMS ONTO STANDARD MERCATOR GRIDS

By

PETER P. K. WONG, CHRISTOPHER GREGORY,

and DAVID W. HANDSCHUMACHER

DECEMBER 1975

Prepared for

OFFICE OF NAVAL RESEARCH

under Contract N00014-75-C-0209,

Project NR 083-603

HAWAII INSTITUTE OF GEOPHYSICS

UNIVERSITY OF HAWAII

Data Report No. 30

HIG-75-19

COMPUTER PROGRAM PROJECTING FICTITIOUS LONGITUDE AND LATITUDE SYSTEMS ONTO STANDARD MERCATOR GRIDS

By

Peter P.K. Wong, Christopher Gregory,
and David W. Handschumacher

See 1473

December 1975

Prepared for
Office of Naval Research
under Contract N00014-75-C-0209,
Project NR 083-603

Approved by Director
George P. Woollard

Date: 30 December 1975

PRECEDING PAGE BLANK-NOT FILMED

ABSTRACT

A computer program has been written in Fortran language that plots fictitious latitudes and longitudes about an arbitrary polar system onto standard Mercator projections. The mathematical development of this program is presented along with a complete program description and program listing. Several example plots generated by the program are included to demonstrate its option characteristics.

TABLE OF CONTENTS

LIST OF FIGURES

INTRODUCTION

This report describes a computer program that plots latitudes and longitudes about any fictitious polar system onto standard Mercator grids of geographic latitudes and longitudes. The principle of this program is shown schematically in Figures 1 and 2. Figure 1A shows the geographic latitude and longitude system for one-half of the earth, 0° to 180° longitude. Figure 1B is the standard Mercator projection at 10° intervals of latitudes and longitudes for 70°N to 70°S latitude, and 0° to 180° longitude. In Figure 2A, one-half of the earth is shown schematically again with geographic latitudes and longitudes, only this time with the addition of a second latitude-longitude system for a second polar pair, P_1 (60°N, 180°) and P_2 (60°S, 0°). In Figure 2B, the fictitious polar system of latitudes and longitudes is shown projected onto a standard Mercator grid.

The mathematical development of the program, presented in the following section, should allow the reader to adapt the program to other computer languages, such as PL/1, Assembly, or Basic. The program description, which includes detailed instructions for the use of job control cards, allows the program to be utilized by workers with limited experience on an IBM/360 computer. A complete program listing is included in the final section of the report, and examples of several plots generated with the program are presented in the Appendix.



Fig. 1. A. Global schematic of standard longitudes for one-half of the earth (0°-180° longitude).
B. Mercator projection for 70°N-70°S latitude and 0°-180° longitude.



Fig. 2. A. Fictitious latitudes-longitudes about fictitious polar system P_1 (60°N, 180°), P_2 (60°S, 0°) on standard latitudes-longitudes of Fig. 1A.

B. Fictitious latitudes-longitudes about P_1, P_2 projected onto standard Mercator piot of Fig. 1B.

Although this program was developed for utilization in geotectonic studies seeking to describe the surface kinematics of the earth's crust (for an example see Figure 15 of Handschumacher, 1976), its applicability to numerous other scientific investigations is apparent. The authors are interested in learning of such application and would appreciate receiving information on how this program was used in other areas of study.

MATHEMATICAL DEVELOPMENT

Consider that the axis of the cylinder tangent to the sphere of radius R points in the direction of the unit vector

$$\underset{\sim}{K} = \underset{\sim}{i} \text{ SIN } \theta_p \text{ COS } \phi_p + \underset{\sim}{j} \text{ SIN } \theta_p \text{ SIN } \phi_p + \underset{\sim}{k} \text{ COS } \theta_p \ , \tag{1}$$

where θ_p is the polar angle of the axis and ϕ_p is the azimuth angle. Here the z-axis is the earth's axis passing through the poles.



Fig. 3.

We may find two other vectors perpendicular to $\underset{\sim}{K}$ and to one another by using an analogy with the spherical coordinate mutually perpendicular vectors. If $\underset{\sim}{K}$ corresponds to $\underset{\sim}{e}_r$ and $\underset{\sim}{J}$ to $\underset{\sim}{e}_\phi$ and $\underset{\sim}{I}$ to $\underset{\sim}{e}_\theta$, we have

$$\underset{\sim}{I} = \underset{\sim}{i} \cos\theta_p \cos\phi_p + \underset{\sim}{j} \cos\theta_p \sin\phi_p - \underset{\sim}{k} \sin\theta_p$$

$$\underset{\sim}{J} = - \underset{\sim}{i} \sin\phi_p + \underset{\sim}{j} \cos\phi_p \qquad (2)$$

$$\underset{\sim}{K} = \underset{\sim}{i} \sin\theta_p \cos\phi_p + \underset{\sim}{j} \sin\theta_p \sin\phi_p + \underset{\sim}{k} \cos\theta_p \quad .$$

$\underset{\sim}{I}$, $\underset{\sim}{J}$, and $\underset{\sim}{K}$ establish another rectangular coordinate system X, Y, and Z as shown in Figure 3.

Every point on the cylinder of radius R with axis Z has position vector

$$\underset{\sim}{r}_c = R \cos\Phi \, \underset{\sim}{I} + R \sin\Phi \, \underset{\sim}{J} + Z \, \underset{\sim}{K} \quad , \qquad (3)$$

where Φ is the azimuth angle in the X, Y, Z system. The radial line with parameter t with azimuth angle Φ the same as equation (3) is

$$\underset{\sim}{r}_M = (\sin\theta \cos\Phi \, \underset{\sim}{I} + \sin\Theta \sin\Phi \, \underset{\sim}{J} + \cos\Theta \, \underset{\sim}{K}) \, t \quad . \qquad (4)$$

Equations (3) and (4) intersect when

$$t \sin\Theta \cos\Phi = R \cos\Phi$$

$$t \sin\Theta \sin\Phi = R \sin\Phi \qquad (4a)$$

$$t \cos\Theta \qquad = Z \quad .$$

Equation (4a) gives us

$$
\begin{aligned}
t \ \mathrm{SIN}\ \theta &= R \\
t \ \mathrm{COS}\ \theta &= Z \\
\text{or} \quad \mathrm{COT}\ \theta &= Z/R \qquad (5) \\
t \quad &= R/\mathrm{SIN}\ \theta
\end{aligned}
$$

Thus from equation (3) and (4) equivalently, we obtain using equation (5)

$$
\begin{aligned}
\underset{\sim}{r}_M &= R/\mathrm{SIN}\ \theta\ (\mathrm{SIN}\ \theta\ \mathrm{COS}\ \Phi\ \underset{\sim}{I} + \mathrm{SIN}\ \theta\ \mathrm{SIN}\ \Phi\ \underset{\sim}{J} + \mathrm{COS}\ \theta\ \underset{\sim}{K}) \\
&= R\ \mathrm{COS}\ \Phi\ \underset{\sim}{I} + R\ \mathrm{SIN}\ \Phi\ \underset{\sim}{J} + \underset{\sim}{K}\ Z \qquad (6) \\
\mathrm{COT}\ \theta &= Z/R\ .
\end{aligned}
$$

Now if we consider the cylinder of radius R whose axis is the z-axis and the radial line with parameter S, we have similar to the above in the x, y, z system

$$
\begin{aligned}
\underset{\sim}{r}_m &= (\mathrm{SIN}\ \theta\ \mathrm{COS}\ \phi\ \underset{\sim}{i} + \mathrm{SIN}\ \theta\ \mathrm{SIN}\ \phi\ \underset{\sim}{j} + \mathrm{COS}\ \theta\ \underset{\sim}{k})S \\
\underset{\sim}{r}_c &= R\ \mathrm{COS}\ \phi\ \underset{\sim}{i} + R\ \mathrm{SIN}\ \phi\ \underset{\sim}{j} + z\ \underset{\sim}{k}\ .
\end{aligned}
\qquad (7)
$$

So, similar to equation (6), the position vector of intersection is:

$$\underset{\sim}{r}_m = R/\text{SIN } \theta \; (\text{SIN } \theta \text{ COS } \phi \; \underset{\sim}{i} + \text{SIN } \theta \text{ SIN } \phi \; \underset{\sim}{j} + \text{COS } \theta \; \underset{\sim}{k})$$

$$= R \text{ COS } \phi \; \underset{\sim}{i} + R \text{ SIN } \phi \; \underset{\sim}{j} + \underset{\sim}{k} \; z \qquad (8)$$

$$\text{COT } \theta = z/R$$

Now $\underset{\sim}{r}_M$ is parallel to $\underset{\sim}{r}_m$ since we are trying to find a correspondence with points on the sphere to its projection on the cylinders by radial lines. Since $\underset{\sim}{r}_M$ is parallel to $\underset{\sim}{r}_m$, we must have

$$\underset{\sim}{r}_M = C \; \underset{\sim}{r}_m \qquad \text{(where C is a constant).} \qquad (9)$$

Thus, $\underset{\sim}{r}_M = C\underset{\sim}{r}_m$. So, from equation (6) and equation (8)

$$\frac{R}{\text{SIN } \Theta} = C \frac{R}{\text{SIN } \theta} \quad \text{or} \quad C = \frac{\text{SIN } \theta}{\text{SIN } \Theta} \; .$$

Therefore, equation (9) may be written

$$\text{SIN } \Theta \; \underset{\sim}{r}_M = \text{SIN } \theta \; \underset{\sim}{r}_m \; . \qquad (10)$$

On introducing the second equations of (6) and (8) in equation (10), we obtain

$$\sin\theta\,(R\cos\Phi\,\underset{\sim}{I} + R\sin\Phi\,\underset{\sim}{J} + \underset{\sim}{K}\,Z)$$
$$= \sin\theta\,(R\cos\phi\,\underset{\sim}{i} + R\sin\phi\,\underset{\sim}{j} + \underset{\sim}{k}\,z)\ . \tag{11}$$

If equation (11) is dotted in turn by $\underset{\sim}{i}$, $\underset{\sim}{j}$, $\underset{\sim}{k}$, with the use of equation (2), we obtain the following:

$$\begin{aligned}
R\sin\theta\cos\phi &= \sin\theta\,(R\cos\Phi\cos\theta_p\cos\phi_p \\
&\quad - R\sin\Phi\sin\phi_p + Z\sin\theta_p\cos\phi_p) \\
R\sin\theta\sin\phi &= \sin\Theta\,(R\cos\Phi\cos\theta_p\sin\phi_p \\
&\quad + R\sin\Phi\cos\phi_p + Z\sin\theta_p\sin\phi_p) \\
z\sin\theta &= \sin\Theta\,(-R\cos\Phi\sin\theta_p + 0 + Z\cos\theta_p)
\end{aligned} \tag{12}$$

If the first two equations above are squared and added together, we obtain

$$\begin{aligned}
\sin\theta = \sin\Theta\,(&\cos^2\Phi\cos^2\theta_p + \sin^2\Phi + (Z/R)^2\sin^2\theta_p \\
&+ 2(Z/R)\cos\Phi\sin\theta_p\cos\theta_p)^{1/2}
\end{aligned} \tag{13}$$

Thus from equation (12) we obtain

$$\cos\phi = \frac{\cos\Phi \cos\theta_p \cos\phi_p - \sin\Phi \sin\phi_p + (Z/R)\sin\theta_p \sin\phi_p}{\sqrt{\cos^2\Phi \cos^2\theta_p + \sin^2\Phi + (Z/R)\sin^2\theta_p + 2(Z/R)\cos\Phi \sin\theta_p \cos\theta_p}}$$

$$\sin\Phi = \frac{\cos\Phi \cos\theta_p \sin\phi_p + \sin\Phi \cos\phi_p + (Z/R)\sin\theta_p \sin\phi_p}{\sqrt{\cos^2\Phi \cos^2\theta_p + \sin^2\Phi + (Z/R)^2 \sin^2\theta_p + 2(Z/R)\cos\Phi \sin\theta_p \cos\theta_p}} \qquad (14)$$

$$z/R = \frac{-\cos\phi \sin\theta_p + (Z/R)\cos\theta_p}{\sqrt{\cos^2\Phi \cos^2\theta_p + \sin^2\Phi + (Z/R)^2 \sin^2\theta_p + 2(Z/R)\cos\Phi \sin\theta_p \cos\theta_p}}$$

Equation (14) enables the (ϕ, z) to be expressed in terms of (Φ, Z).

Let us make the following conventions:



Fig. 4.

where θ_p = latitude of fictitious pole referring to True North Pole

ϕ_p = longitude of the fictitious pole referring to True North Pole

ϕ_p = 0 passes through Greenwich

$\theta_p = 0$ North Pole

$\theta_p = \pi$ South Pole

is measured from west to east.

Thus 53°W has for ϕ_p, $\phi_p = 360° - 53°$.

Also 53°N corresponds to $\theta_p = 90° - 53°$.

Notice that if the fictitious pole is south of the equator then 53°S would correspond to $\theta_p = 90° + 53°$.

The simplest way to generate the map of fictitious longitude and latitude curves on the Mercator map is to use equation (14).

Define

$$A \equiv \cos\Phi \cos\theta_p \cos\phi_p - \sin\Phi \sin\phi_p + \cot\Theta \sin\theta_p \cos\phi_p$$

$$B \equiv \cos\Phi \cos\theta_p \sin\phi_p + \sin\Phi \cos\phi_p + \cot\Theta \sin\theta_p \sin\phi_p \tag{15}$$

$$R \equiv \sqrt{\cos^2\Phi \cos^2\theta_p + \sin^2\Phi + \cot^2\Theta \sin^2\theta_p + 2\cot\Theta \cos\Phi \sin\theta_p \cos\theta_p}$$

$$\equiv \sqrt{A^2 + B^2}$$

where we have put $Z/R = \cot\Theta$.

To generate the maps of the fictitious longitudes put $\Phi = \phi_p$ into (15) where ϕ_p is varied over the fictitious meridians.

Then from equation (14) we have

$$\phi_p = \mathrm{TAN}^{-1}(B/A) \quad \text{or} \quad \mathrm{SIN}^{-1}(B/(A^2 + B^2)^{1/2})$$

$$\mathrm{COT}\ \theta = z/R = \frac{-\ \mathrm{COS}\ \phi_p\ \mathrm{SIN}\ \theta_p + \mathrm{COT}\ \theta\ \mathrm{COS}\ \theta_p}{\sqrt{A^2 + B^2}} \tag{16}$$

and vary θ through $0 < \theta < \pi$ to generate the map. Equation (16) is the parametric equation for the map with parameter being θ.

To generate the maps of the fictitious latitudes we put $\theta = \theta_p$ into equation (15) above and let Φ range through $0 < \Phi < 2\pi$ to generate the maps corresponding to each θ_p.

Note that if $\mathrm{Tan}^{-1}\frac{B}{A}$ is used we must put

$$\phi = \mathrm{TAN}^{-1}(B/A) + [1 - H(A)] * \pi + H(A) * [1 - H(B)] * 2\pi$$

where

$$H(X) = 0 \quad \text{if } X < 0$$

$$= 1 \quad \text{if } X \geq 0$$

However, if the above convention is not used, the following can be considered:

$$\phi = \mathrm{TAN}^{-1}(B/A) \leq 0 \qquad \text{WEST}$$

$$\phi = \mathrm{TAN}^{-1}(B/A) > 0 \qquad \text{EAST}$$

PROGRAM DESCRIPTION

The program listed in this report is written in Fortran IV at a sufficiently generalized level to be compatible with or easily adapted to most computers and plotters. It has been compiled and executed successfully on IBM 370 and IBM 360 computer systems, and the plotting has been tested on Calcomp and XYNETICS plotters as well. The program is subdivided into a Main program, and three subroutines, FICT, EDGEPT, DEGREE, and also a function program YMER.

The function of the Main program is to enter the input data. It computes the required constants, converts the coordinates, and checks the map for the size in Y-axis. It calls in subroutine DEGREE to develop latitude-longitude annotation at the left and the bottom for the Mercator map. The arrangement of data cards and various options are discussed in the section on data input preparation. Once the Mercator map has been completed, it calls the subroutine FICT to generate all the fictitious curves on that particular Mercator map, and last, it either looks for another set of data for Pole projection or terminates its operation.

Subroutine EDGEPT is called by the subroutine FICT to extrapolate the fictitious curve to the edge of the Mercator map if a data point is being found outside the map.

Function program YMER is called by the Main program and subroutine FICT to calculate the displacement from the equator to transform the point found on a sphere into a point as if on an ellipsoid.

Dimensions

The program is dimensioned so that fictitious longitude and fictitious latitude curves are obtainable for up to 160 parameters. This number is arbitrary; however, it may be made larger by increasing the dimensions for the appropriate program variable, FPLON. No other changes in the program are necessary.

If one wants to process additional plots in one execution time, another set of input cards is necessary and should be included, as shown below, after the comment card of the previous set of data.

```
//MERPLOT JOB (9999, 1M, 3KI, 3KL), PETER
// EXEC FORTCLG
//SYSIN DD *
          ----FORTRAN SOURCE PROGRAM----
//GO.PLOTTAPE DD UNIT=(7TRK,,DEFER),LABEL=(,NL),
// VOL=SER=PLOT,DISP=(NEW,KEEP)
//GO.SYSIN DD *
   40.00N    110.00W    10    10
50N    50S    70W    130E    .063    10    1    1
POLE PROJECTION    40N    110W
/*
```

Minor modifications to the program are necessary for a successful compilation and execution on the other computers. Statements like CALL PLOTS, CALL PLOT, CALL SYMBOL, have to be changed if the plotter used is not the Calcomp or XYNETICS plotter. The size of Y-maximum can be increased more than the 10-inch limit if the allowable size on Y-axis is bigger than that of the 11-inch Calcomp plotter.

Input Preparation and Options

CARD 1	READS IN LOCATION OF THE POLE AND THE FICTITIOUS GRID SPACING (DEGREE/INTERVAL)	
COL. 1-5	LATITUDE OF POLE	F5.2
COL. 6	NORTH OR SOUTH (N OR S)	A1
COL. 8-13	LONGITUDE OF POLE	F6.2
COL. 14	EAST OR WEST (E OR W)	A1
COL. 15-17	GRID SPACING IN FICTITIOUS LONGITUDE CURVES	F3.0
COL. 18-20	GRID SPACING IN FICTITIOUS LATITUDES	F3.0

CARD 2	READS IN THE SIZE OF THE MAP	
COL. 1-5	TOP OF THE MERCATOR MAP	F5.0
COL. 6	NORTH OR SOUTH (N OR S)	A1
COL. 8-12	BOTTOM OF THE MERCATOR MAP	F5.0
COL. 13	NORTH OR SOUTH (N OR S)	A1
COL. 15-19	RIGHT LONGITUDE OF THE MERCATOR MAP	F5.0

COL. 21	EAST OR WEST (E OR W)	A1
COL. 22-26	LEFT LONGITUDE OF THE MERCATOR MAP	F5.0
COL. 27	EAST OR WEST INFORMATION (E OR W)	A1
COL. 29-34	MERCATOR SCALE (INCH/DEGREE)	F6.3
COL. 36-41	MERCATOR GRID SPACING (DEGREE/INTERVAL)	F6.3
COL. 55	IF=1 FICTITIOUS LONGITUDES ARE DRAWN ON THE MERCATOR MAP	I1
COL. 59	IF=1 FICTITIOUS LATITUDES ARE DRAWN ON THE MAP	I1

CARD 3

COL. 1-64	TITLE OF THE PLOT	16A4

PROGRAM LISTING

```
C  ********************************************************************
C  *                                                                  *
C  *    THIS PROGRAM IS TO PLOT FICTITIOUS MERIDIANS AND PARALLELS ON *
C  *               REGULAR MERCATOR.   *** NOV. 11 1974. ***          *
C  *                     HAWAII INSTITUTE OF GEOPHYSICS               *
C  *                                                                  *
C  ********************************************************************
C
      COMMON XMT,XMB,YMT,YMB,YTOP,YBOT,SSC,XRIGH,IST(7)
      INTEGER A,B,C,D,N/'N'/,E/'E'/
      DIMENSION BUFF(1000),TITLE(16)
C
C*****INITIALIZE THE PLOT
C
C*****CALL PLOTS(1,1,1) IS INITIAL CALL FOR XYNETICS PLOTTER
C*****CALL PLOTS(BUFF,4000) IS INITIAL CALL FOR CALCOMP PLOTTER
C
      CALL PLOTS(BUFF,4000)
      XPOT=0.
      YAX=0.
   14 READ (5,8,END=7) FLAT,NS,FLON,EW,SCLEN,SCLAT
    8 FORMAT(F5.2,A1,1X,F6.2,A1,F3.0,F3.0)
      READ (5,10,END=7)YT,A,YB,B,XR,C,XL,D,SCALE,GRID,IST
   10 FORMAT(4(F5.0,A1,1X),2(F6.3,1X),10X,7I1)
      READ(5,9,END=7)TITLE
    9 FORMAT(16A4)
      PRINT 11,YT,A,YB,B,XR,C,XL,D,SCALE,GRID
   11 FORMAT('1','  PLOT LIMITS:   ',4(F7.3,A1,2X)//,2X,' SCALE=',F7.3,
     &' INCHES/DEGREE'//,3X,'GRID SPACING =',F7.3,' DEGREES/INTERVAL'//)
      PRINT 15,IST
   15 FORMAT('0','X',' 1  2  3  4  5  6  7'/
     &7X,7(I1,2X)/
     * 6X,'53 54 55 56 57 58 59')
      PRINT 12,TITLE
   12 FORMAT(///5X,16A4//)
C
C*****PRINT MAP TITLE AND SET THE PEN TO ORIGIN.
C
      CALL SYMBOL(XPOT+2.5,YAX,.14,TITLE,0.,64)
      CALL PLOT(XPOT+2.0,YAX+0.75,-3)
      SSC=SCALE/60.
      IF(A.EQ.N.AND.B.EQ.N) GO TO 1
      IF(A.NE.N.AND.B.NE.N) GO TO 2
      YMT=YT*60
      YMB=-YB*60
      GO TO 3
    1 YMT=YT*60
      YMB=YB*60
      GO TO 3
    2 YMT=-YT*60
      YMB=-YB*60
    3 IF(C.EQ.E.AND.D.EQ.E) GO TO 4
      IF(C.NE.E.AND.D.NE.E) GO TO 5
      XMT=10800.-XR*60
      XMB=XL*60-10800.
      GO TO 6
    4 XMT=XR*60-10800.
      XMB=XL*60-10800.
      GO TO 6
    5 XMT=10800.-XR*60
      XMB=10800.-XL*60
    6 YBOT=YMER(YMB)
      YTOP=(YMER(YMT)-YBOT)*SSC
C
C*****CHECK THE OVERALL SIZE OF THE MAP.Y-MAX. IS 10 INCH.
C*****THERE IS NO X-MAX. ON THE CALCOMP PLOTTER TECHNICALLY.
C
      IF(YTOP.GE.10.) GO TO 33
      XRIGH=(XMT-XMB)*SSC
      PRINT 17, XRIGH,YTOP
   17 FORMAT(/5X,' MAP SIZE IN INCHES',
     &' X:',F7.3,5X,'Y:',F7.3////)
      NXA=(XMT-XMB)/(60.*GRID)+1.5
      XM=(XMT-XMB)/(NXA-1)
      DO 21 I=1,NXA
      XPOT=XM*(I-1)*SSC
      XEW=XMB+(I-1)*GRID*60.
      IF(MOD(I,2).EQ.0) GO TO 77
C
C*****DRAW VERTICAL GRIDS AND ANNOTATE IN DEGREES.
C
      CALL DEGREE(XPOT,XEW,1)
      CALL PLOT(XPOT,0.,3)
      CALL PLOT(XPOT,YTOP,2)
      GO TO 21
   77 CALL PLOT(XPOT,YTOP,3)
      CALL PLOT(XPOT,0.,2)
C
```

```
C*****IF SCALE TOO SMALL AND GRIDS TOO CLOSE EVERY OTHER
C*****LONGITUDINAL GRID IS ANNOTATED TO AVOID INVISIBILITY
C
      IF(SCALE.LT.0.06.AND.GRID.LE.10.) GO TO 21
      CALL DEGREE(XPOT,XEW,1)
   21 CONTINUE
      NYA=(YMT-YMB)/(60.*GRID)+1.5
      YM=(YMT-YMB)/(NYA-1)
      DO 22 I=1,NYA
      YPOT=(YMER(YMB+YM*(I-1))-YBOT)*SSC
      YNS=YMB+(I-1)*GRID*60.
      IF(MOD(I,2).EQ.0) GO TO 88
C
C*****DRAW HORIZONTAL GRIDS AND ANNOTATE IN DEGREES.
C
      CALL PLOT(XRIGH,YPOT,3)
      CALL PLOT(0.,YPOT,2)
      CALL DEGREE(YPOT,YNS,2)
      GO TO 22
   88 CALL DEGREE(YPOT,YNS,2)
      CALL PLOT(0.,YPOT,3)
      CALL PLOT(XRIGH,YPOT,2)
   22 CONTINUE
C
C*****IF CARD 2. COL. 55 = 1. FICTITIOUS LONGITUDE CURVES ARE DRAWN.
C*****OTHERWISE ONLY REGULAR MERCATOR MAP IS PROVIDED.
C
      IF(IST(3).NE.0) CALL FICT(FLAT,NS,FLON,FW,SCLON,SCLAT)
      YAX=-0.75
      GO TO 14
   33 PRINT 55,YTOP
   55 FORMAT(///' SCALE/SIZE ERROR. PLOT TOO LARGE VERTICALLY. YTOP=',
     &F7.3,' INCHES, MAX = 10.0 INCHES.')
C
C*****CALL END PLOT TO CLOSE THE ROUTINE
C
    7 CALL PLOT(0.,0.,999)
      STOP
      END



                SUBROUTINE FICT(FLAT,NS,FLON,FW,SCLON,SCLAT)
C
C ***************************************************************
C *                                                             *
C *   THIS SUBROUTINE IS TO CALCULATE THE FICTITIOUS MERIDIANS AND *
C *              PARALLELS SUPERIMPOSED ON THE MERCATOR.        *
C *                                                             *
C ***************************************************************
C
      COMMON XMT,XMB,YMT,YMB,YTOP,YBOT,SSC,XRIGH,IST(7)
      DIMENSION NST(2),EWT(4),FPLON(160)
      INTEGER N/'N'/,E/'E'/,NS,FW,NST/'N','S'/,EWT/'E','W','N','S'/
      INTEGER SN,FLAG,EWF,IO/0/,KP/0/
      REAL RAD/.01745329/,DEG/57.2957795/,PI/3.141592654/,HPI/1.5707963/
      REAL LON,LAT
      IF((XMT.EQ.10800.).AND.(XMB.EQ.0.)) KP=1
      IF((XMT.GE.9600.).AND.(XMB.LE.-9600.)) KP=2
      PRINT 8,FLAT,NS,FLON,FW
    8 FORMAT(1H+,5X,'FICTITIOUS POLE = ',F5.2,A1,3X,F6.2,A1///)
      IF(NS.EQ.N) GO TO 2
      FLAT=90.+FLAT
      GO TO 3
    2 FLAT=90.-FLAT
    3 IF(EW.NE.E) FLON=360.-FLON
      FLATR=RAD*FLAT
      FLONR=RAD*FLON
C
C*****NO. OF POINTS GENERATED IN BOTH LONGITUDINAL AND
C*****LATITUDINAL DIRECTIONS
C
      NLON=(360./SCLON)
      NLAT=(175./SCLAT)
      CH=SCLON*60.*2.
      CPLON=-SCLON
      FLAG=0
  200 SCLAT=SCLAT/2.
      LO=NLON
      LA=NLAT*2+1
      IF(FLAG.EQ.1) CPLON=0.
      DO 10 I=1,LO
      CPLON=CPLON+SCLON
      FPLON(I)=CPLON
C
C*****FICTITIOUS LATITUDE CURVES ARE DRAWN WHEN FLAG=1
C
      IF(FLAG.EQ.1) GO TO 203
      IF(FPLON(I).GE.180.) GO TO 38
      EWF=EWT(1)
      GO TO 39
```

```
   38 FWF=FWT(3)
      FPLON(I)=360.-FPLON(I)
      GO TO 39
  203 IF(FPLON(I).GT.90.) GO TO 205
      FPLON(I)=90.-FPLON(I)
      FWF=FWT(1)
      GO TO 39
  205 FWF=FWT(4)
      FPLON(I)=FPLON(I)-90.
   39 CPLONR=RAD*CPLON
      CPLAT=0.
      IF(FLAG.EQ.1) CPLAT=-SCLAT
      KKK=0
      KK=0
      K=0
      NW=0
      DO 20 J=1,LA
      KKK=KKK+1
      IF(FLAG.EQ.1) KK=KK+1
      CPLAT=CPLAT+SCLAT
      CPLATR=RAD*CPLAT
      A=CPLONR
      IF(FLAG.EQ.1) A=CPLATR
      B=CPLATR
      IF(FLAG.EQ.1) B=CPLONR
C
C*****EQUATIONS USED FOR GENERATING FICTITIOUS POINTS IN THE MERCATOR
C
      A1=COS(A)*COS(FLATR)*COS(FLONR)
      A2=SIN(A)*SIN(FLONR)
      A3=COTAN(B)*SIN(FLATR)*COS(FLONR)
      AA=A1-A2+A3
      B1=COS(A)*COS(FLATR)*SIN(FLONR)
      B2=SIN(A)*COS(FLONR)
      B3=COTAN(B)*SIN(FLATR)*SIN(FLONR)
      BB=B1+B2+B3
      C=SQRT(AA**2+BB**2)
      D=(-COS(A)*SIN(FLATR)+COTAN(B)*COS(FLATR))
      LAT=ATAN2(C,D)
      LON=ATAN2(BB,AA)
C
C*****CALCULATE THE LATITUDE OF POINT Q IN MERCATOR SYSTEM
C
      IF(LAT.GE.HPI) GO TO 35
      LAT=HPI-LAT
      ALAT=LAT*60.*DEG
      SN=NST(1)
      GO TO 36
   35 LAT=LAT-HPI
      ALAT=-LAT*60.*DEG
      SN=NST(2)
   36 BLAT=LAT*DEG
C
C*****CALCULATE THE LONGITUDE OF POINT Q IN MERCATOR SYSTEM
C
      IF(LON.GT.0.) GO TO 37
      FW=FWT(2)
      BLON=ABS(LON)*DEG
      ALON=10800.-BLON*60.
      GO TO 12
   37 EW=FWT(1)
      BLON=LON*DEG
      ALON=BLON*60.-10800.
   12 IF(J.GT.1) GO TO 13
      ALOS1=ALON
      ALAS1=ALAT
   13 ALOS2=ALOS1
      ALAS2=ALAS1
      ALOS1=ALON
      ALAS1=ALAT
      FPT=XMT-ALOS2
      FPI=XMB-ALOS2
      IF(K.GT.1) GO TO 16
C
C*****CHECK IF POINT IS OUTSIDE THE MERCATOR MAP
C
      IF(ALAT.GT.YMT.OR.ALAT.LT.YMB) GO TO 21
      IF(ALON.GT.XMT.OR.ALON.LT.XMB) GO TO 21
      IF(J.EQ.1) GO TO 5
      IF (K.GE.1) GO TO 16
C
C*****EXTRAPOLATE THE CURVE TO THE EDGE IF DATA POINT IS FOUND
C*****OUTSIDE THE MAP
C
C*****SPECIAL CASE WHEN PLOTTING IS HANDLED FROM 0W TO 0E OR 0W TO 180W
C
      IF(KP.EQ.0) GO TO 71
      IF(KP.GE.1.AND.NW.EQ.1) GO TO 72
```

```
      IF(ALOS2.GE.0..AND.ALOS2.LE.XMT.AND.ALON.LE.XMT.AND.ALON.GE.0.)GO
     &TO 71
      IF(ALOS2.LE.0..AND.ALOS2.GE.XMB.AND.ALON.LE.0..AND.ALON.GE.XMB)GO
     &TO 71
      IF( ABS(ALOS2).LE.CH.AND.ABS(ALON).LE.CH) GO TO 71
      IF(ALOS2.GE.XMB.AND.ALON.GT.0..AND.ALOS2.LT.0.)ALOS2=XMT
      IF(KP.EQ.2.AND.ALOS2.LE.XMT.AND.ALON.LT.0.)ALOS2=XMB
      GO TO 71
   72 IF(ALOS2.EQ.XMT.OR.ALOS2.EQ.XMB) NW=0
      IF(ALOS2.EQ.XMT) GO TO 77
      IF(ALOS2.EQ.XMB) ALOS2=XMT
      GO TO 71
   77 ALOS2=XMB
   71 IF(ALAS2.GE.YMT) CALL EDGEPT(ALAS2,ALOS2,ALAT,ALON,PLAS,PLOS,1)
      IF(ALOS2.LE.XMB) CALL EDGEPT(ALAS2,ALOS2,ALAT,ALON,PLAS,PLOS,2)
      IF(ALAS2.LE.YMB) CALL EDGEPT(ALAS2,ALOS2,ALAT,ALON,PLAS,PLOS,1)
      IF(ALOS2.GE.XMT) CALL EDGEPT(ALAS2,ALOS2,ALAT,ALON,PLAS,PLOS,2)
      K=K+1
      ALON1=(PLOS-XMB)*SSC
      ALAT1=(YMER(PLAS)-YBOT)*SSC
      CALL PLOT(ALON1,ALAT1,3)
      GO TO 5
C
C*****CHECK IF POINT IS OUTSIDE THE MERCATOR MAP
C
   16 IF(ALAT.GT.YMT.OR.ALAT.LT.YMB) GO TO 21
      IF(ALON.GT.XMT.OR.ALON.LT.XMB) GO TO 21
C
C*****SPECIAL CASE MERCATOR PLOT FROM OW TO OE
C
      IF(KP.EQ.2.AND.EPT.LE.CH.AND.ALON.GE.XMB.AND.ALON.LT.0.)ALON=XMT
      IF(KP.EQ.2.AND.EPT.GE.-CH.AND.ALON.GT.0..AND.EPT.LT.0.)ALON=XMB
      IF(ALON.EQ.XMT)ALOS1=XMT
      IF(ALON.EQ.XMB)ALOS1=XMB
    5 K=K+1
      IO=IO+1
      IF(FLAG.NE.1.AND.IO.EQ.1) PRINT 17,FPLON(I),EWF
      IF(FLAG.EQ.1.AND.IO.EQ.1) PRINT 18,FPLON(I),EWF
   18 FORMAT(5X,'FICTITIOUS LATITUDE ',F6.2,A1,' HAS THE FOLLOWING DATA
     &POINTS:-'/)
   17 FORMAT(5X,'FICTITIOUS LONGITUDE ',F6.2,A1,' HAS THE FOLLOWING DATA
     & POINTS:-'/)
      PRINT 11,BLAT,SN,BLON,EW
   11 FORMAT(20X,' LATITUDE = ',F6.2,A1,10X,' LONGITUDE = ',F6.2,A1)
      ALON1=(ALON-XMB)*SSC
      ALAT1=(YMER(ALAT)-YBOT)*SSC
      IC=2
      IF(J.EQ.1) IC=3
      CALL PLOT(ALON1,ALAT1,IC)
      IF(KP.EQ.2.AND.(ALON.EQ.XMT.OR.ALON.EQ.XMB)) K=0
      IF(KP.EQ.2.AND.(ALON.EQ.XMT.OR.ALON.EQ.XMB))NW=1
      GO TO 20
   21 IF(K) 20,20,6
   20 CONTINUE
      PRINT 115
  115 FORMAT('/')
      IO=0
      GO TO 10
C
C*****EXTRAPOLATE THE CURVE TO THE EDGE IF DATA POINT IS FOUND
C*****OUTSIDE THE MAP
C
    6 IF(KP.EQ.1.AND.ALON.EQ.XMT) GO TO 23
      IF(KP.EQ.1.AND.EPT.LE.CH.AND.ALON.LT.XMB) ALON=XMT
      IF(KP.EQ.2.AND.EPT.LE.CH.AND.ALON.GE.XMB.AND.ALON.LT.0.)ALON=XMT
      IF(KP.EQ.2.AND.EPT.GE.-CH.AND.ALON.GT.0..AND.EPT.LT.0.)ALON=XMB
      IF(ALAT.GE.YMT) CALL EDGEPT(ALAT,ALON,ALAS2,ALOS2,PLAS,PLOS,1)
      IF(ALON.GE.XMT) CALL EDGEPT(ALAT,ALON,ALAS2,ALOS2,PLAS,PLOS,2)
      IF(ALAT.LE.YMB) CALL EDGEPT(ALAT,ALON,ALAS2,ALOS2,PLAS,PLOS,1)
      IF(ALON.LE.XMB) CALL EDGEPT(ALAT,ALON,ALAS2,ALOS2,PLAS,PLOS,2)
      PRINT 112
  112 FORMAT('/')
      ALON1=(PLOS-XMB)*SSC
      ALAT1=(YMER(PLAS)-YBOT)*SSC
      CALL PLOT(ALON1,ALAT1,2)
   23 K=0
      IF(KK.LT.LA.OR.KKK.LT.LA) GO TO 20
   10 CONTINUE
C
C*****IF CARD 2 COL.59 NE. 0.FICTITIOUS LATITUDE CURVES ARE DRAWN.
C*****OTHERWISE RETURN TO MAIN PROGRAM.
C
      IF(IST(7).EQ.0) GO TO 15
      FLAG=FLAG+1
      IF(FLAG.GT.1) GO TO 15
      PNLON=NLON
      NLON=NLAT
```

```
      NLAT=PNLON
      SCLAT=SCLAT*2.
      TSCLAT=SCLON
      SCLON=SCLAT
      SCLAT=TSCLAT
      PRINT 112
      GO TO 200
   15 RETURN
      END


      SUBROUTINE EDGEPT(ALAS2,ALOS2,ALAT,ALON,PLAS,PLOS,N)
C *********************************************************************
C *                                                                   *
C *    EDGEPT IS TO EXTRAPOLATE THE FICTITIOUS CURVES TO THE EDGE OF  *
C *    THE MAP IF THERE IS ANOTHER POINT GENERATED OUTSIDE THE MAP.   *
C *                                                                   *
C *********************************************************************
C
      COMMON XMT,XMB,YMT,YMB,YTOP,YBOT,SSC,XRIGH,IST(7)
      AB=ABS(ALON-ALOS2)
      CD=ABS(ALAS2-ALAT)
      IF(N.GT.1) GO TO 3
      IF(ALOS2-XMT) 102,102,103
  102 IF(ALOS2-XMB) 103,104,104
  104 PLAS=YMT
      IF(ALAS2.LE.YMB) PLAS=YMB
      PLOS=ALOS2+(ALON-ALOS2)*(ALAS2-PLAS)/(ALAS2-ALAT)
      GO TO 17
  103 IF(AB.GE.CD) GO TO 109
      GO TO 104
    3 IF(ALAS2-YMT) 106,106,107
  106 IF(ALAS2-YMB) 107,109,109
  109 PLOS=XMT
      IF(ALOS2.LE.XMB) PLOS=XMB
      PLAS=ALAS2+(ALAT-ALAS2)*(ALOS2-PLOS)/(ALOS2-ALON)
      GO TO 17
  107 IF(AB.GE.CD) GO TO 109
      GO TO 104
   17 RETURN
      END


      SUBROUTINE DEGREE(XYLOC,XY,NO)
C
C *********************************************************************
C *                                                                   *
C *      DEGREE DEVELOPS LAT-LON ANNOTATION FOR THE MERCATOR MAP.     *
C *                                                                   *
C *                                                                   *
C *********************************************************************
C
      INTEGER*2      A,N/'N'/,S/'S'/,W/'W'/,E/'E'/
      REAL*4 XYLOC,XY,DEG
      IF(NO.EQ.2) GO TO 10
      IF(XY.LT.0.) GO TO 1
      DEG=(10800.-XY)/60.
      A=W
      GO TO 6
    1 DEG=(10800.+XY)/60.
      A=E
      GO TO 6
   10 IF(XY.LT.0) GO TO 2
      DEG=XY/60.
      A=N
      GO TO 7
    2 DEG=-XY/60.
      A=S
      GO TO 7
    6 CALL NUMBER(XYLOC,-.25,0.07,DEG,0.,0)
      CALL SYMBOL(999.,-.25,0.07,A,0.0,2)
      RETURN
    7 CALL NUMBER(-.4,XYLOC,0.07,DEG,0.,0)
      CALL SYMBOL(999.,XYLOC,0.07,A,0.0,2)
      RETURN
      END
```

```
                    FUNCTION YMER(Y)
C
C  ***********************************************************************
C  *                                                                     *
C  *       YMER CALCULATES THE DISPLACEMENT FROM THE EQUATOR FOR         *
C  *            A MERCATOR PROJECTION OF AN INPUT IN MINUTES             *
C  *                                                                     *
C  ***********************************************************************
C
      REAL*8 RA,P4/.785398163397448/
      RA=Y
      RA=RA*.0002908882
      YMER=7915.704468*DLOG10(DTAN(P4+RA*.5))
     *-23.268932*DSIN(RA)-.0525*DSIN(RA)**3
     1-.000213*DSIN(RA)**5
      RETURN
      END
```

PLOT LIMITS: 70.000N 70.000S 0.0 W 0.0 E

SCALE= 0.032 INCHES/DEGREE

GRID SPACING = 10.000 DEGREES/INTERVAL

1	2	3	4	5	6	7
0	0	1	0	0	0	1
53	54	55	56	57	58	59

POLE PROJECTION 60N 180W

MAP SIZE IN INCHES X: 11.520 Y: 6.340

FICTITIOUS POLE = 60.00N 180.00W

FICTITIOUS LONGITUDE 0.0 E HAS THE FOLLOWING DATA POINTS:-

LATITUDE =	55.00N	LONGITUDE =	180.00E
LATITUDE =	50.00N	LONGITUDE =	180.00E
LATITUDE =	45.00N	LONGITUDE =	180.00E
LATITUDE =	40.00N	LONGITUDE =	180.00E
LATITUDE =	35.00N	LONGITUDE =	180.00E
LATITUDE =	30.00N	LONGITUDE =	180.00E
LATITUDE =	25.00N	LONGITUDE =	180.00E
LATITUDE =	20.00N	LONGITUDE =	180.00E
LATITUDE =	15.00N	LONGITUDE =	180.00E
LATITUDE =	10.00N	LONGITUDE =	180.00E
LATITUDE =	5.00N	LONGITUDE =	180.00E
LATITUDE =	0.00S	LONGITUDE =	180.00E
LATITUDE =	5.00S	LONGITUDE =	180.00E
LATITUDE =	10.00S	LONGITUDE =	180.00E
LATITUDE =	15.00S	LONGITUDE =	180.00E
LATITUDE =	20.00S	LONGITUDE =	180.00E
LATITUDE =	25.00S	LONGITUDE =	180.00E
LATITUDE =	30.00S	LONGITUDE =	180.00E
LATITUDE =	35.00S	LONGITUDE =	180.00E
LATITUDE =	40.00S	LONGITUDE =	180.00E
LATITUDE =	45.00S	LONGITUDE =	180.00E
LATITUDE =	50.00S	LONGITUDE =	180.00E
LATITUDE =	55.00S	LONGITUDE =	180.00E
LATITUDE =	60.00S	LONGITUDE =	180.00E
LATITUDE =	65.00S	LONGITUDE =	180.00E
LATITUDE =	65.00S	LONGITUDE =	0.00W

FICTITIOUS LONGITUDE 10.00E HAS THE FOLLOWING DATA POINTS:-

LATITUDE =	55.07N	LONGITUDE =	178.49W
LATITUDE =	50.12N	LONGITUDE =	177.30W
LATITUDE =	45.16N	LONGITUDE =	176.35W
LATITUDE =	40.19N	LONGITUDE =	175.54W
LATITUDE =	35.22N	LONGITUDE =	174.85W
LATITUDE =	30.25N	LONGITUDE =	174.23W
LATITUDE =	25.28N	LONGITUDE =	173.68W
LATITUDE =	20.30N	LONGITUDE =	173.17W
LATITUDE =	15.32N	LONGITUDE =	172.69W
LATITUDE =	10.34N	LONGITUDE =	172.23W
LATITUDE =	5.36N	LONGITUDE =	171.79W
LATITUDE =	0.38N	LONGITUDE =	171.35W
LATITUDE =	4.60S	LONGITUDE =	170.92W
LATITUDE =	9.59S	LONGITUDE =	170.47W
LATITUDE =	14.57S	LONGITUDE =	170.02W
LATITUDE =	19.55S	LONGITUDE =	169.54W
LATITUDE =	24.52S	LONGITUDE =	169.04W
LATITUDE =	29.50S	LONGITUDE =	168.49W
LATITUDE =	34.47S	LONGITUDE =	167.82W
LATITUDE =	39.44S	LONGITUDE =	167.21W
LATITUDE =	44.41S	LONGITUDE =	166.42W
LATITUDE =	49.37S	LONGITUDE =	165.49W
LATITUDE =	54.32S	LONGITUDE =	164.35W
LATITUDE =	59.26S	LONGITUDE =	162.89W
LATITUDE =	64.17S	LONGITUDE =	160.95W
LATITUDE =	69.05S	LONGITUDE =	158.16W
LATITUDE =	69.78S	LONGITUDE =	5.01W
LATITUDE =	64.91S	LONGITUDE =	2.05W

FICTITIOUS LONGITUDE 20.00E HAS THE FOLLOWING DATA POINTS:-

LATITUDE = 55.26N	LONGITUDE = 177.00W
LATITUDE = 50.47N	LONGITUDE = 174.65W
LATITUDE = 45.64N	LONGITUDE = 172.73W
LATITUDE = 40.78N	LONGITUDE = 171.11W
LATITUDE = 35.90N	LONGITUDE = 169.72W
LATITUDE = 31.00N	LONGITUDE = 168.49W
LATITUDE = 26.10N	LONGITUDE = 167.38W
LATITUDE = 21.19N	LONGITUDE = 166.36W
LATITUDE = 16.27N	LONGITUDE = 165.41W
LATITUDE = 11.35N	LONGITUDE = 164.50W
LATITUDE = 6.42N	LONGITUDE = 163.62W
LATITUDE = 1.50N	LONGITUDE = 162.76W
LATITUDE = 3.43S	LONGITUDE = 161.91W
LATITUDE = 8.36S	LONGITUDE = 161.04W
LATITUDE = 13.28S	LONGITUDE = 160.16W
LATITUDE = 18.20S	LONGITUDE = 159.23W
LATITUDE = 23.12S	LONGITUDE = 158.26W
LATITUDE = 28.02S	LONGITUDE = 157.20W
LATITUDE = 32.93S	LONGITUDE = 156.05W
LATITUDE = 37.81S	LONGITUDE = 154.76W
LATITUDE = 42.69S	LONGITUDE = 153.29W
LATITUDE = 47.54S	LONGITUDE = 151.57W
LATITUDE = 52.36S	LONGITUDE = 149.50W
LATITUDE = 57.13S	LONGITUDE = 146.92W
LATITUDE = 61.84S	LONGITUDE = 143.59W
LATITUDE = 66.43S	LONGITUDE = 139.06W
LATITUDE = 69.14S	LONGITUDE = 9.60W
LATITUDE = 64.65S	LONGITUDE = 3.99W

ACKNOWLEDGMENTS

The authors thank Rita Pujalet for editorial assistance and Jean Taga for manuscript preparation.

This work was funded by the Office of Naval Research under contract N00014-75-C-0209.

REFERENCES

Handschumacher, David, 1976, Post-Eocene plate tectonics of the eastern Pacific, in The Geophysics of the Pacific Ocean Basin and its Margin: A Volume in Honor of George P. Woollard, Geophys. Monogr. Ser., vol. 19, edited by George H. Sutton, Murli H. Manghnani, and Ralph Moberly, AGU, Washington, D.C. (in press).

PRECEDING PAGE BLANK-NOT FILMED

A P P E N D I X

Sample Plots



POLE 60N 180W FICT



POLE 60N 180W FICT



POLE 70N 145W FICT



POLE 40N 110W FICT



PØLE 50N 170W FICT



FICT. PØLE ON 80W

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER
HIG-75-19, Data Report #30

2. GOVT ACCESSION NO

3. RECIPIENT'S CATALOG NUMBER

4. TITLE (and Subtitle)
Computer Program Projecting Fictitious Longitude and Latitude Systems onto Standard Mercator Grids.

5. TYPE OF REPORT & PERIOD COVERED
Data Report No. 30

6. PERFORMING ORG. REPORT NUMBER
(14) HIG-75-19, HIG-Data-30

7. AUTHOR(s)
(10) Peter P.K. Wong, Christopher Gregory, David W. Handschumacher

(12) 31p.

8. CONTRACT OR GRANT NUMBER(s)
(15) N00014-75-C-0209

9. PERFORMING ORGANIZATION NAME AND ADDRESS
Hawaii Institute of Geophysics
2525 Correa Road
Honolulu, HI 96822

10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS
(16) NR-083-603

11. CONTROLLING OFFICE NAME AND ADDRESS
Office of Naval Research (Code 480)
Ocean Science & Technology Division
Arlington, VA 22217

12. REPORT DATE
November 1975

13. NUMBER OF PAGES

14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)
(11) Dec 75

15. SECURITY CLASS. (of this report)
Unclassified

15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Cartography

Computer program

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

A computer program has been written in Fortran language that plots fictitious latitudes and longitudes about an arbitrary polar system onto standard Mercator projections. The mathematical development of this program is presented along with a complete program description and program listing. Several example plots generated by the program are included to demonstrate its option characteristics.

MANDATORY DISTRIBUTION LIST

FOR UNCLASSIFIED TECHNICAL REPORTS, REPRINTS, & FINAL REPORTS
PUBLISHED BY OCEANOGRAPHIC CONTRACTORS
OF THE OCEAN SCIENCE AND TECHNOLOGY DIVISION
OF THE OFFICE OF NAVAL RESEARCH
(REVISED JAN 1975)

1 Director of Defense Research
and Engineering
Office of the Secretary of Defense
Washington, D. C. 20301
ATTN: Office, Assistant Director
(Research)

Office of Naval Research
Arlington, Virginia 22217
3 ATTN: (Code 480)*
1 ATTN: (Code 460)
1 ATTN: (Code 102-OS)
6 ATTN: (Code 102IP)

1 ONR ResRep (if any)

Director
Naval Research Laboratory
Washington, D. C. 20375
6 ATTN: Library, Code 2620

12** Defense Documentation
Center
Cameron Station
Alexandria, Virginia
22314

Commander
Naval Oceanographic
Office
Washington, D. C. 20390
1 ATTN: Code 1640
1 ATTN: Code 70

1 NODC/NOAA
Rockville, MD 20882

TOTAL REQUIRED - 35 copies

* Add one separate copy of
Form DD-1473

** Send with these 12 copies
two completed forms DDC-50,
one self addressed back to
contractor, the other addressed to ONR, Code 480